
ABSTRACT
Most ongoing clinical trials have a data monitoring committee and/or an internally motivated sponsor who
periodically reviews the accumulating safety data of their enrolled subjects. Subject-level listings and
patient profiles are essential aids in the review of clinical data. To ease the burden on the data reviewers,
sponsors increasingly request color coding outputs in order to easily identify old/deleted records from the
previous data transfer and highlight new records from the current data transfer. At PharmaStat, we have
developed a SAS macro, TrackCHG, to meet this increasingly popular request for color-coded subject
level listings and patient profiles. This paper explains the components and functions of the TrackCHG
SAS macro within the context of generating of an adverse event listing. The provided sample datasets
and SAS code allow the reader to experiment and implement the TrackCHG macro into existing listing
generation programs and produce their own color-coded outputs.

INTRODUCTION
Most clinical trials have an independent data monitoring committee (DMC) who reviews accumulating
data from clinical trials on an ongoing basis. In order to provide quality data to the DMC, sponsors
conduct an internal review of available data prior to each DMC meeting. Ongoing clinical data reviews
rely on subject-level listings and patient profiles in conjunction with data management activities. This
review task becomes onerous as the amount of data accumulates and the number of data transfers
increases.
Color-coded subject-level listings are one of the tools to reduce the workload and increase the efficiency
for ongoing clinical data review. Color-coded records highlight the differences between two versions of
the data and allow reviewers to focus on the changes between the versions. We have developed a SAS
macro, TrackCHG, that can be easily integrated with existing listing generation programs to
programmatically apply customized text formats to records.
First, we present how the TrackCHG macro fits into the workflow of creating subject-level listings. Then,
we walk through the SAS macro code as it performs the following functions: 1) records the date stamp of
the current data, 2) uses a PROC SORT step and DATA MERGE step to compare the previously derived
listing dataset against the current derived listing dataset, 3) apply a text format to color code the new and
old records in a customizable fashion. Finally, we piece all the steps together to produce an adverse
event listing that has color-coded purple text to identify old data from the initial data cut on January 1,
2022, and blue text to identify new records from today’s data transfer.

WORKFLOW OF CREATING A COLOR-CODED SUBJECT-LEVEL LISTING
Figure 1 shows the workflow to generate a non-color-coded subject-level listing output at PharmaStat.
The source data for the listing can come from any number of sources including but not limited to: raw
electronic data capture (EDC) datasets, clinical data management (CDM) datasets, study data tabulation
model (SDTM) datasets, or analysis data model (ADaM) datasets. The listing program reads in the source
data, processes the data and formats the records into a nice and presentable structure, thus creating two
outputs: a derived listing SAS dataset, henceforth referred to as AFILE, which is read by the PROC
REPORT procedure to also generate a listing output (Figure 2).

Figure 1

Flow Chart of the Listing Generation Process

Source Data
(Raw data, CDM, SDTM, ADaM)

Listing
Program

Derived
Listing Data

Non-color Coded
Listing Output
Figure 2

Sample Listing of Adverse Events

In order to generate color-coded subject-level listings, several steps need to be taken in order to retrieve
a former version of the derived listing dataset, perform a comparison between the current and former
derived listing, and color code the differences. This process has been consolidated into the TrackCHG
SAS macro, so that this macro call can be easily incorporated inside a listing generation program.
Figure 3 shows the process and Figure 6 displays the colorized output. The listing created from PROC
REPORT can be produced in a number of outputs including: HTML, portable document format (PDF), rich
text format (RTF), Microsoft Word, and Microsoft Excel. The TrackCHG macro applies text formatting that
is compatible with all of these document types.

Figure 3

Flow Chart of a Color-Coded Listing Generation Process with TrackCHG

TRACKCHG COMPONENTS AND FUNCTION
The TrackCHG macro performs the following functions to produce color-coded output: 1) records the date
of the current data transfer in a permanent dataset, 2) performs a full outer join to compare the records of
the previously derived listing dataset against the records in the current derived listing dataset records,
3) programmatically applies a customizable text format to color code the new and old records. In order to
perform these functions the TrackCHG macro uses three input datasets and seven macro parameters to
produce three output datasets.

TRACKCHG INPUTS THREE DATASETS
1. TLGDATA.DOTsaved: a dataset that stores the date stamps of data transfers. It contains 4
variables:
Table 1 DOTSaved Variables
Variable Name

Variable Label

Type

Format

Length

DataDate

Date of Current Data Transfer

Integer YYMMDD10.

8

PrevDataDate

Date of Previous Data Transfer

Integer YYMMDD10.

8

TimeStamp

Date Stamp of Current Data Transfer

Char

$9

9

PrevStamp

Date Stamp of Previous Data Transfer

Char

$9

9

Note: This dataset can also be configured to store other variables, for example, the run date of the listing,
or the date when the source SDTM or ADaM dataset was updated.


Date of Current Data Transfer (integer, date), This is the date last modified of the dataset
specified within the TrackCHG macro. In the sample code (Appendix A: TrackCHG.SAS), the
SAS macro grabs the date last modified from the Adverse Events Analysis dataset (Appendix B3:
ADAE (Adverse Event Analysis Dataset)), but I recommend modifying the macro code to grab the
date that the raw data or SDTM demographics dataset was last updated to better capture the
date when the files were transferred.



Date of Previous Data Transfer (integer, date), This is the date when the previous version of data
files was transferred, derived using the lag function in a DATA step.



Date Stamp of Current Data Transfer (character, $9) This is the character value of the Date of
Current Data Transfer, preceded by an underscore. These values follow the format
“_YYYYMMDD”

Date Stamp of Previous Data Transfer (character, $9) This the character value of the Date of
Previous Data Transfer, preceded by an underscore. These values follow the format
“_YYYYMMDD”

Using the sample SAS code in Appendix B1: DOTsaved (Date of Transfer Saved), the
TLGDATA.DOTsaved dataset can be created with the initial value of the first observation with a
TimeStamp = ‘NONE’, and a second record containing the time stamp of January 1, 2022, the date of the
first data cut we received from this clinical study.
Figure 4

DOTsaved Date of Transfer Dataset

2. AFILE: the derived dataset that supports the listing, which is produced from the current data
transfer. This is the dataset that is usually read by the PROC REPORT procedure, but instead will
be input and processed by the TrackCHG macro. In the sample code, AFILE is derived from the
adverse event analysis dataset (Appendix B3: ADAE (Adverse Event Analysis Dataset)) in the
listing code, Appendix C: Sample Adverse Event Listing Code.
3. TIMEDATA.&BASEPROG.&ORDER.&PREVSTAMP: a derived listing dataset that supports the
listing, which was produced from the previous data transfer. The sample SAS code in
Appendix B2: L_AE_SAMP_20220101 (Sample Adverse Event Listing Derived Dataset) ,
provides L_AE_SAMP_20220101, a derived listing dataset for adverse events with an artificially
assigned date stamp of January 1, 2022 can be created with initial values (see image below for
example).
Figure 5

Derived Adverse Event Listing Data Set with Date Stamp 2022-01-01

TRACKCHG HAS SEVEN MACRO PARAMETERS
The first three parameters (&INDATA, &BASEPROG, &ORDER) are used to identify, save, and retrieve
the appropriately named derived listing datasets. The following two parameters (&NEWCOLOR and
4

&PREVCOLOR) allow the users to easily customize the text formats applied to identify new and old
records, and &ARRAYVAR specifies which character variables—corresponding to the columns in the
output listings--will have the text formatting applied to them. &SORTVAR is used for data comparison
when merging the derived listing dataset with its predecessor.
Table 2 TrackCHG Macro Parameters
Valid
Value

Parameter

Sample Value

Description

1.

INDATA

Dataset
name

INDATA = AFILE

2.

BASEPROG

text

BASEPROG = L_AE

The name of the base listing program
The order of the listing program, may have values to
indicate the subsets or variations of the base
programs

Input dataset name, the derived dataset that
supports the listing to be generated

3.

ORDER

text

ORDER = _SAMP

4.

NEWCOLOR

text

NEWCOLOR =
^S={color=blue}

The text format to be applied to new/changed data
from the current data transfer. The default is set to
blue font color

5.

PREVCOLOR

text

PREVCOLOR =
^S={textdecoration=
line_through
color=purple}

The text format to be applied original/deleted data
from the previous data transfer. The default is set to
a purple font color with strikethrough

6.

SORTVAR

Variable
names

SORTVAR = subjid
aeseq subjid_c date_c
aecat_c aestat_c
aesevrel_c aeacn_c

The list of variable names, delimited by a space, for
the sort and merging order of AFILE with its
timestamped predecessor. This list typically contains
all the variables that are used in the COLUMN
statement in PROC REPORT

7.

ARRAYVAR

Character
variable
names

ARRAYVAR = subjid_c
date_c aecat_c
aestat_c aesevrel_c
aeacn_c

The list of variable names, delimited by a space, for
the character variables whose values will be
formatted with either the new or previous text
formatting

TRACKCHG OUTPUTS THREE DATASETS
1. TLGDATA.DOTsaved: a dataset that stores the date stamps of data transfers, updated to include
the current date stamp.
2. work._&BASEPROG.&ORDER: a derived dataset that supports the listing output, which has
colored text formatting applied to new and old records. This is the dataset that will be read into
the PROC REPORT procedure in place of its predecessor, AFILE. The TrackCHG macro applies
the following changes to work._&BASEPROG.&ORDER:
Table 3 work._&BASEPROG.&ORDER Created Variables
Variable Name

Variable Label

Type

Format

Length

NewDataFL

New Data Record Flag

Char

$1

1

ColorData

Apply Color/Text Formatting to this Record?

Char

$3

3

Data_Color

Color/Text Formatting

Char

$50

50



New Data Record Flag (character, $1), This is set to ‘Y’ if a record in AFILE has new data, null
otherwise. This flag can be used in PROC REPORT or additional data steps for extra processing
of new data.



Apply Color/Text Formatting to this Record? (character, $3), This is set to ‘YES’ to identify that
the character variables identified in &ARRAYVAR should be assigned color/text formatting.

Color/Text Formatting (character, $50) This variable is assigned values from &NEWCOLOR or
&PREVCOLOR, as applicable. For records where ColorData = ‘YES’, all of the variables
identified in &ARRAYVAR will have the text formatting specified by Data_Color appended to the
beginning of their values.



TIMEDATA.&BASEPROG.&ORDER.&TIMESTAMP: AFILE saved in the TimeData folder with a
date stamp of the current data transfer. This dataset will be used for future comparisons.

TRACKCHG MACRO FUNCTION
This macro is designed to be called by a listing program, right before the PROC REPORT procedure. The
derived listing dataset, AFILE, that is normally read in by the PROC REPORT will instead be processed
by the TrackCHG macro, which will compare the records from AFILE to the records generated from a
previous iteration of the listing program. TrackCHG will apply text formats to character variables in AFILE
to color-code changes in the data.
This macro consists two parts: time stamping and data comparison.

CREATING TIME STAMPED VARIABLES
In order for the TrackCHG macro to run with the sample data provided in Appendix C: Sample Adverse
Event Listing Code, the macro is currently designed to go into the ADaM derived data directory to retrieve
a date (e.g., the date of the ADAE.sas7bdat file was last updated). This date is saved in the macro
variable, RAWDATE:
%let
%let
%let
%let

rc
fid
rawdate
fidc

=
=
=
=

%sysfunc(filename(onefile, &adamdata\adae.sas7bdat)) ;
%sysfunc(fopen(&onefile)) ;
%substr(%qsysfunc(finfo(&fid,Last Modified)),1,9) ;
%sysfunc(fclose(&fid)) ;

The first DATA step creates a temporary dataset, NewSaved, with one observation and two variables
DATADATE, which is a numeric variable containing the SAS date of RAWDATE, and TIMESTAMP which
is the character variable storing digits of RAWDATE preceded by an underscore. Additionally, the macro
variable TIMESTAMP is created in this step and will be used to save date stamped versions of AFILE, the
derived listing dataset.
data NewSaved ;
length TimeStamp $9. ;
DataDate = input("&rawdate", date9.) ;
TimeStamp = '_' || compress(put(DataDate,yymmdd10.), '-') ;
call symput('timestamp', timestamp);
format datadate yymmdd10.;
run ;

The second DATA step is designed to record the unique values of DATADATE in DOTsaved. Datasets
WORK.NewSaved and TLGDATA.DOTsaved are merged by DATADATE in a full outer join. If
DATADATE is a new date that is not contained in the TLGDATA.DOTsaved, it creates a new record in
TLGDATA.DOTsaved. If the DATADATE already exists in TLGDATA.DOTsaved, then no changes are
made. In this data step, we also define the two new variables PREVDATADATE and PREVSTAMP, which
holds the YYMMDD10. and character values, respectively, of the previously stored value of DATADATE.
These two macro variables will aid in the comparison and saving of date stamped derived datasets. The
value of PREVSTAMP is also stored in the macro variable, PREVSTAMP, in order to be able to call the
derived listing made from the previous date:
/* Record date of current data transfer in tlgdata.DOTsaved */
data tlgdata.DOTsaved ;
merge tlgdata.DOTsaved (in=SavedDates)
newSaved ;
by dataDate ;
lagPrevDataDate = lag(DataDate) ;
lagPrevStamp = lag(TimeStamp) ;
if not SavedDates then do ;
PrevDataDate = lagprevDataDate ;
PrevStamp = lagprevStamp ;

end ;
call symput('prevstamp',prevstamp);
drop lagprevStamp lagprevDataDate ;
format DataDate PrevDataDate YYMMDD10. ;
run ;

COMPARING TWO VERSIONS OF DATA
The second part of the macro indicates new or changed data within the derived dataset that will be used
to create the listing. A data saves AFILE with the timestamp of TIMESTAMP and stores the dataset in the
TimeData folder for future comparisons:
data timedata.&baseprog.&order.&timestamp ;
set afile ;
run ;

If there is no previous record (&PREVSTAMP= NONE) then the macro passes the dataset through
without adding any color text formats to the data because the previous version of the dataset does not
exist and would cause the MERGE step to fail.
Otherwise, after saving AFILE, the previous (TIMEDATA.&BASEPROG.&ORDER.&PREVSTAMP)
version of the derived listing dataset and the current version (AFILE) of the derived listing dataset are
merged by the variables listed in &SORTVAR in an full outer join. Three new variables, NewDataFL,
ColorData, and Data_Color, are defined (Table 3) in this data step as well. Any records that do not fall
within the inner join of the two datasets will be marked for text formatting (ColorData = ‘YES’). If the
records are not in the previously derived dataset, then it must be “new” data, and is flagged NewDataFL =
‘Y’ and marked for text formatting ColorData = ‘YES’, and the Data_Color is set to &NEWCOLOR. If the
records are not in the new data then they must have been changed or removed. In this case, the previous
data records are also marked for text formatting ColorData = ‘YES’ and the Data_Color is set to the
&PREVCOLOR. Then, a DO Loop is called and goes through the array of character variables defined in
&ARRAYVAR and it appends the text formatting at the beginning of the character value of each of the
array variables, when ColorData = ‘YES’:
data _&baseprog.&order ;
length &arrayvar $250 data_color $50;
%if &prevstamp = NONE %then %do ;
set &indata ;
array cols{*} &arrayvar ;
%end ;
%else %do ;
merge timedata.&baseprog.&order.&prevstamp (in=inprev)
&indata (in=innew) ;
by &sortvar ;
if not inprev then do ;
newdatafl = 'Y' ;
colordata = 'YES' ;
data_color = "&newcolor" ;
end ;
/* if not in new data, then be removed/changed records in previous data */
if not innew then do ;
colordata = 'YES' ;
data_color = "&prevcolor" ;
end ;
if colordata='YES' then do ;
do iy = 1 to dim(cols) ;
cols{iy} = strip(data_color) || strip(cols{iy}) ;
end ;
end ;
%end ;
drop iy;
run ;

The text formatting is appended to the beginning of the values of the variables specified by the macro
parameter &ARRAYVAR, and these values are stored in a temporary dataset _&BASEPROG.&ORDER.
Once this temporary dataset is read by PROC REPORT, the colorized adverse event listing shown in
Figure 6 is produced.

COLOR-CODED LISTING OUTPUT
Figure 6 demonstrates how the colored output can be used to identify changed, new, and deleted
records. The third adverse event for Subject 001-001 can be identified as a changed record because
there is a new record in blue text immediately followed by the previous record in purple strikethrough text
that describe the same adverse event. The previous record shows that the foot fracture did not have an
end datetime and had an outcome of “Recovering or Resolving” because the event was ongoing at the
time of the previous data transfer, 2022-01-01. As of the current data transfer, the foot fracture is
“Recovered or Resolved” and has a populated end date, end time, and study day. Subject 003-003 has a
new adverse event of COVID-19 pneumonia, displayed in blue text. Subject 004-004 had a second AE of
weight decreased that was removed because this record is not in the current data transfer but was
included in the previous data transfer.
Figure 6

Sample Color-Coded Listing of Adverse Events

The text formatting can be customized for old/changed data and new/current data. These text formats
include strikethrough, highlights, bold, italics, colors, and more. Please refer to SAS documentation,
SAS® 9.4 Output Delivery System: Advanced Topic and Eric Gerbhart’s SAS Global Forum Paper 2222009, on SAS text formats.
If the previous old/changed records are not the review focus, the previous records can easily be
suppressed by implementing the following SAS code:
/* if not in previous data, then must be new data */
if not inprev then do ;
newdatafl = 'Y' ;
colordata = 'YES' ;
data_color = "&newcolor" ;
end ;
/* if not in new data, then must be removed/changed records in previous data */
if not innew then delete ;

If there is a set of listings that are always produced and delivered to the Sponsor or DMC following each
data cut, this set of listings can all use the same DOTsaved dataset, instead of creating a unique
TLGDATA.DOTsaved dataset to be used by each listing program. It is important to note that all the listing
programs associated with a TLGDATA.DOTsaved dataset should be run at least once per data transfer in
order to log and store the timestamped derived listing datasets of all the listings from each data transfer.

CAUTIONARY TALES
One weakness of this TrackCHG macro is that it is not flexible to changes in the derived dataset that
supports the listing.
If, for example, a sponsor wants to add a new column like a flag for COVID-related events, the macro will
crash. Workaround: Update the PROC REPORT to include the new variable, but do not include the new
variable name in the &SORTVAR and &ARRAYVAR values until the next data transfer, when the new
variable becomes available in the previous and current datasets.
If you modify one or more of the values in the listings (e.g., displaying AETERM in proper case instead of
uppercase), all affected observations will be flagged as new records. Workaround: implement these
updates prior to the data transfer.

CONCLUSION
With only minor updates to the environment and listing macros, the add-on TrackCHG macro can be
easily implemented to generate colorized listings that highlight the differences between two sets of source
data. Modifications can be made to the macro to customize the text formats to add strikethrough, bold,
italicize, highlighting, etc. These text formats also can be carried over to Microsoft Excel listings. A SAS
user can also preferentially display only new records, only old records, or both in the color-coded subject
listings.

APPENDIX A: TRACKCHG.SAS
**
**
**
**
**
**
**
**
**
**
**
**
**
**
**
**
**
**
**

--------------------------------------------------------------------- ** ;
Program:
TrackCHG.sas
;
Author:
IMLeprince
;
Created:
2021-11-15
;
Inputs:
tlgdata.DOTsaved
Contains timestamps of data transfers
;
timedata.l_baseprog&order.&prevstamp
AFILE created from
;
previous (&prevstamp) data transfer
;
work.afile
Analysis file for listing
;
;
Outputs:
tlgdata.DOTsaved
Contains timestamps of data transfers
;
timedata.l_baseprog&order.&timestamp
AFILE saved with a
;
timestamp
;
work._&baseprog.&order
Analysis file for listing with
;
color-coded records
;
;
Usage:
Track Changes between current vs previous data transfer
;
for subject-level safety listings by color coding records ;
Revisions:
;
--------------------------------------------------------------------- ** ;

%macro TrackCHG (
indata
=
baseprog =
order
=
newcolor =
prevcolor =
sortvar
=
arrayvar =
) ;

,
,
,
^S={color=blue},
^S={textdecoration=line_through color=purple},
,

%*-------------------------------------------------------------------------;
%* Save most current data transfer date into DOTsaved
;
%* Create macro variables:
;
%* (1) Timestamp: _YYYYMMDD of current data transfer
;
%* (2) PrevStamp: _YYYYMMDD of previous data transfer
;
%*-------------------------------------------------------------------------;
/* Preference to get the date of data transfer from the raw demographics dataset: DM.sas7bdat */
/* For utilization with example AE listing, getting date of data transfer from ADaM ADAE.sas7bdat */
%let rc
= %sysfunc(filename(onefile, &adamdata\adae.sas7bdat)) ;
%let fid
= %sysfunc(fopen(&onefile)) ;
%let rawdate = %substr(%qsysfunc(finfo(&fid,Last Modified)),1,9) ;
%let fidc
= %sysfunc(fclose(&fid)) ;
data NewSaved ;
length TimeStamp $9. ;
DataDate = input("&rawdate", date9.) ;
TimeStamp = '_' || compress(put(DataDate,yymmdd10.), '-') ;

call symput('timestamp', timestamp);
run ;
%put &=timestamp ;
/* Record date of current data transfer in tlgdata.DOTsaved */
data tlgdata.DOTsaved ;
merge tlgdata.DOTsaved (in=SavedDates)
newSaved ;
by dataDate ;
lagPrevDataDate = lag(DataDate) ;
lagPrevStamp = lag(TimeStamp) ;
if not SavedDates then do ;
PrevDataDate = lagprevDataDate ;
PrevStamp = lagprevStamp ;
end ;
call symput('prevstamp',prevstamp);
drop lagprevStamp lagprevDataDate ;
format DataDate PrevDataDate YYMMDD10. ;
run ;
%*--------------------------------------------------------------------------;
%* Indicate New or Changed Data within Listing
%*--------------------------------------------------------------------------;
/* Sort input data by &SORTVAR */
proc sort data = &indata ;
by &sortvar ;
run ;
/* Save a timestamped copy of the current dataset for future comparison */
data timedata.&baseprog.&order.&timestamp ;
set &indata ;
run ;
/* Merge previous and current derived datasets
*/
/* Previous data transfer: timedata.&baseprog.&order.&prevstamp */
/* Current data transfer: &indata
*/
data _&baseprog.&order ;
length &arrayvar $250 data_color $50;
%if &prevstamp = NONE %then %do ;
set &indata ;
array cols{*} &arrayvar ;
%end ;
%else %do ;
merge timedata.&baseprog.&order.&prevstamp (in=inprev)
&indata (in=innew) ;
by &sortvar ;

12

array cols{*} &arrayvar ;
if not inprev then do ;
newdatafl = 'Y' ;
colordata = 'YES' ;
data_color = "&newcolor" ;
end ;
/* if not in new data, then must be removed/changed records in previous data */
if not innew then do ;
colordata = 'YES' ;
data_color = "&prevcolor" ;
end ;+
if colordata='YES' then do ;
do iy = 1 to dim(cols) ;
cols{iy} = strip(data_color) || strip(cols{iy}) ;
end ;
end ;
%end ;
run ;
%mend ;

APPENDIX B: SAMPLE DATA
%*-------------------------------------------------------------------------;
%* Setting up environment: define the file paths
%*-------------------------------------------------------------------------;
/*
/*
/*
/*
/*

ADAMDATA = where the ADaM datasets (ADAE) resides
TLGDATA = where the derived datasets that support outputs resides
TIMEDATA = where the timestamped derived datasets that support
outputs resides
OUTPUTS = where the AE listing.RTF will be saved

%let
%let
%let
%let

ADAMDATA = ..\CSR\ADaM\Derived Data ;
TLGDATA = ..\CSR\TLGs\Derived Data ;
TIMEDATA = ..\CSR\TLGs\Derived Data\TimeStamped ;
OUTPUTS = ..\CSR\TLGs\Outputs ;

libname
libname
libname
libname

tlgdata "&tlgdata" ;
timedata "&timedata" ;
adamdata "&adamdata" ;
outputs "&outputs" ;

Figure 7

Sample Environment

*/
*/
*/
*/
*/

ADAMDATA

TLGDATA
TIMEDATA
OUTPUTS

APPENDIX B1: DOTSAVED (DATE OF TRANSFER SAVED)
%*-------------------------------------------------------------------------;
%* DOTsaved: Date of Transfer Saved
%* Stores the date stamps of data transfers
%*-------------------------------------------------------------------------;
data tlgdata.DOTsaved ;
infile datalines missover ;
informat TimeStamp PrevStamp $9. DataDate PrevDataDate YYMMDD10. ;
input TimeStamp PrevStamp $ DataDate PrevDataDate ;
label TimeStamp = "Date Stamp of Current Data Transfer"
PrevStamp = "Date Stamp of Previous Data Transfer"
DataDate = "Date of Current Data Transfer"
PrevDataDate = "Date of Previous Data Transfer";
format DataDate PrevDataDate YYMMDD10. ;
datalines ;
NONE
_20220101 NONE
2022-01-01
;
run ;

APPENDIX B2: L_AE_SAMP_20220101 (SAMPLE ADVERSE EVENT LISTING DERIVED DATASET)
%*-------------------------------------------------------------------------;
%* l_ae_samp_20220101: derived dataset of AE listing from the
%*
2022-01-01 data transfer
%*-------------------------------------------------------------------------;
data timedata.l_ae_samp_20220101 ;
infile datalines dlm = '*' missover ;
input SUBJID $7. AESEQ :8. SUBJID_C :$7. DATE_C :$50. AECAT_C :$125. AESTAT_C :$15.
AESEVREL_C :$80. AEACN_C :$25. ;
label SUBJID = 'Subject ID'
AESEQ = 'Sequence Number'
SUBJID_C = 'Subejct ID (C)'
DATE_C
= 'Onset Date/End Date (C)'
AECAT_C = 'SOC/PT/Verbatim Term (C)'
AESTAT_C = 'TEAE/SAE/Special Interest (C)'
AESEVREL_C = 'Severity/Relationship to Study Drug/Outcome (C)'
AEACN_C = 'Action Taken with Study Drug (C)'
;
datalines ;
001-001*1*001-001*2018-06-22 17:00 (25) /^n2018-06-22 17:00 (25)*Nervous system disorders^n^R"\b "Headache^R"\b0 "^n^R"\i
"HEADACHE^R"\i0"
*Yes/^nNo/^nNo *Grade: 1/^n^R"\b "Not Related^R"\b0 "/^n(Recovered or
Resolved)
*Dose Not Changed

001-001*2*001-001*2018-09-19 (114) /^n2018-09-22 (117)
*Metabolism and nutrition disorders^n^R"\b "Hypoglycaemia^R"\b0
"^n^R"\i "HYPOGLYCEMIA^R"\i0"
*Yes/^nYes/^nNo *Grade: 3/^n^R"\b "Not Related^R"\b0 "/^n(Recovered or Resolved with
Sequelae)*Dose Not Changed
001-001*3*001-001*2018-09-30 15:18 (125) /^nOngoing
*Injury, poisoning and procedural complications^n^R"\b "Foot
fracture^R"\b0 "^n^R"\i "FRACTURE OF GREAT TOE^R"\i0"*Yes/^nNo/^nNo *Grade: 1/^n^R"\b "Not Related^R"\b0 "/^n(Recovering or
Resolving)
*Not Applicable
001-001*4*001-001*2018-10-10 (135) /^nOngoing
*Metabolism and nutrition disorders^n^R"\b "Type 2 diabetes
mellitus^R"\b0 "^n^R"\i "DIABETES TYPE II^R"\i0"
*Yes/^nNo/^nNo *Grade: 1/^n^R"\b "Not Related^R"\b0 "/^n(Not Recovered or Not
Resolved)
*Dose Not Changed
002-002*1*002-002*2018-03-11 (67) /^nOngoing
*Vascular disorders^n^R"\b "Hypertension^R"\b0 "^n^R"\i "HIGH BLOOD
PRESSURE^R"\i0"
*Yes/^nNo/^nNo *Grade: 1/^n^R"\b "Not Related^R"\b0 "/^n(Not Recovered or Not Resolved)
*Dose Not Changed
002-002*2*002-002*2018-04-11 9:56 (98) /^nOngoing
*Cardiac disorders^n^R"\b "Atrial fibrillation^R"\b0 "^n^R"\i "ATRIAL
FIBRILATION^R"\i0"
*Yes/^nNo/^nNo *Grade: 2/^n^R"\b "Not Related^R"\b0 "/^n(Not Recovered or Not Resolved)
*Dose Not Changed
004-004*1*004-004*2018-04-25 7:45 (115) /^n2018-04-25 7:45 (115)*Gastrointestinal disorders^n^R"\b "Dyspepsia^R"\b0 "^n^R"\i
"HEARTBURN^R"\i0"
*Yes/^nNo/^nNo *Grade: 1/^n^R"\b "Not Related^R"\b0 "/^n(Recovered or Resolved
with Sequelae)*Dose Not Changed
004-004*2*004-004*2019-02-17 (334) /^n2019-03-10 (355)
*Investigations^n^R"\b "Weight decreased^R"\b0 "^n^R"\i "WEIGHT
LOSS^R"\i0"
*Yes/^nNo/^nYes *Grade: 2/^n^R"\b "Not Related^R"\b0 "/^n(Recovered or Resolved)
*Dose Not Changed
;
run ;

APPENDIX B3: ADAE (ADVERSE EVENT ANALYSIS DATASET)
%*-------------------------------------------------------------------------;
%* ADAE: analysis dataset of adverse events
%* Current data transfer
%*-------------------------------------------------------------------------;
data adamdata.adae ;
infile datalines dlm = '*' missover ;
input SUBJID $7. AESEQ :8. ASTDT MMDDYY10. ASTTM :time5. ASTDY :8. AENDT MMDDYY10. AENTM :time5. AENDY :8.
AETERM :$25. AEDECOD :$24. AEBODSYS :$50. AEOUT :$35. ATOXGR :$8. AEREL :$11. AEACN :$16.
TRTEMFL $1. AESER $1. AESIFL $1. ;
label SUBJID
= 'Subject Identifier'
AESEQ
= 'Sequence Number'
ASTDT
= 'Analysis Start Date'
ASTTM
= 'Analysis Start Time'
ASTDY
= 'Analysis Start Relative Day'
AENDT
= 'Analysis End Date'
AENTM
= 'Analysis End Time'
AENDY
= 'Analysis End Relative Day'
AETERM
= 'Reported Term for the Adverse Event'
AEDECOD = 'Dictionary-Derived Term'
AEBODSYS = 'Body System or Organ Class'
AEOUT
= 'Outcome of Adverse Event'
ATOXGR
= 'Analysis Toxicity Grade'
AEREL
= 'Causality'
AEACN
= 'Action Taken with Study Treatment'
TRTEMFL = 'Treatment Emergent Analysis Flag'
AESER
= 'Serious Event'
AESIFL
= 'AE of Special Interest Flag' ;
format ASTDT AENDT YYMMDD10. ASTTM AENTM TIME5. ;
datalines ;
001-001*1*06/22/2018*17:00* 25*06/22/2018*23:00* 25*HEADACHE
*Headache
*Nervous system disorders
*RECOVERED OR RESOLVED
*Grade: 1*NOT RELATED*DOSE NOT CHANGED*YNN
001-001*2*09/19/2018*
*114*09/22/2018*
*117*HYPOGLYCEMIA
*Hypoglycaemia
*Metabolism and nutrition
disorders
*RECOVERED OR RESOLVED WITH SEQUELAE*Grade: 3*NOT RELATED*DOSE NOT CHANGED*YYN
001-001*3*09/30/2018*15:18*125*12/04/2018*
*190*FRACTURE OF GREAT TOE
*Foot fracture
*Injury, poisoning and
procedural complications*RECOVERED OR RESOLVED
*Grade: 1*NOT RELATED*NOT APPLICABLE *YNN
001-001*4*10/10/2018*
*135*
*
*
*DIABETES TYPE II
*Type 2 diabetes mellitus*Metabolism and nutrition
disorders
*NOT RECOVERED OR NOT RESOLVED
*Grade: 1*NOT RELATED*DOSE NOT CHANGED*YNN
002-002*1*03/11/2018*
* 67*
*
*
*HIGH BLOOD PRESSURE
*Hypertension
*Vascular disorders
*NOT RECOVERED OR NOT RESOLVED
*Grade: 1*NOT RELATED*DOSE NOT CHANGED*YNN
002-002*2*04/11/2018*09:56* 98*
*
*
*ATRIAL FIBRILATION
*Atrial fibrillation
*Cardiac disorders
*NOT RECOVERED OR NOT RESOLVED
*Grade: 2*NOT RELATED*DOSE NOT CHANGED*YNN
003-003*1*02/05/2018*03:30* 36*03/13/2018*
* 72*PNEUMONIA DUE TO COVID-19*COVID-19 pneumonia
*Infections and infestations
*FATAL
*Grade: 5*NOT RELATED*DRUG WITHDRAWN *YYN
004-004*1*04/25/2018*07:45*115*04/25/2018*12:30*115*HEARTBURN
*Dyspepsia
*Gastrointestinal disorders
*RECOVERED OR RESOLVED WITH SEQUELAE*Grade: 1*NOT RELATED*DOSE NOT CHANGED*YNN
;
run ;

APPENDIX C: SAMPLE ADVERSE EVENT LISTING CODE
**
**
**
**
**
**
**
**
**
**
**
**
**
**
**
**

--------------------------------------------------Program:
l_ae.sas
Author:
IMLeprince
Created:
2021-12-24

**
**
**
**
**
Usage:
Generate a listings of Adverse Events
**
**
REQUIRES rawdata:
**
REQUIRES anadata: adae
**
REQUIRES macros: TrackCHG
**
**
OUTPUT: l_ae&order.rtf
**
**
Revisions:
**
**
--------------------------------------------------- **

;
;
;
;
;
;
;
;
;
;
;
;
;
;
;
;

%*-------------------------------------------------------------------------;
%* Setting up environment: define the filepaths
%* Refer to Figure 7
%*-------------------------------------------------------------------------;
%let ORDER = _SAMP ;
/*
/*
/*
/*
/*

ADAMDATA = where the ADaM datasets (ADAE) resides
TLGDATA = where the derived datasets that support outputs resides
TIMEDATA = where the timestamped derived datasets that support
outputs resides
OUTPUTS = where the AE listing.RTF will be saved

%let
%let
%let
%let

ADAMDATA = ..\CSR\ADaM\Derived Data ;
TLGDATA = ..\CSR\TLGs\Derived Data ;
TIMEDATA = ..\CSR\TLGs\Derived Data\TimeStamped ;
OUTPUTS = ..\CSR\TLGs\Outputs ;

libname
libname
libname
libname

tlgdata "&tlgdata" ;
timedata "&timedata" ;
adamdata "&adamdata" ;
outputs "&outputs" ;

proc format ;
value $ yn
'Y' = 'Yes'
'N' = 'No'
' ' = 'No'
;
value $ grade
'Grade 1' = 'Grade 1: Mild'
'Grade 2' = 'Grade 2: Moderate'
'Grade 3' = 'Grade 3: Severe'

*/
*/
*/
*/
*/

'Grade 4' = 'Grade 4: Life-threatening'
'Grade 5' = 'Grade 5: Death'
;
value $ aeout
'NOT RECOVERED OR NOT RESOLVED' = 'Not Recovered or Not Resolved'
'RECOVERED OR RESOLVED'
= 'Recovered or Resolved'
'RECOVERING OR RESOLVING'
= 'Recovering or Resolving'
'RECOVERED OR RESOLVED WITH SEQUELAE' = 'Recovered or Resolved with Sequelae'
'UNKNOWN'
= 'Unknown'
'FATAL'
= 'Fatal'
;
run ;
%*-----------------------------------------------------------------------------;
%* Bring in data ADAE
%*-----------------------------------------------------------------------------;
proc sort data = adamdata.adae out = file1 ;
by subjid astdt aendt aebodsys ;
run;
data afile ;
length subjid_c $50
stdate_c endate_c date_c aecat_c aestat_c aesevrel_c aeacn_c $250 ;
set file1 ;
by subjid astdt aendt aebodsys ;
subjid_c = subjid ;
/* process character dates */
array daten {*} astdt aendt ;
array times {*} asttm aettm ;
array dates_c {*} $50 stdate_c endate_c ;
array adys
{*} astdy aendy ;
do ii = 1 to dim(daten) ;
if adys{ii} > .z then do ;
if asttm > .z then
dates_c{ii} = strip(put(daten{ii},yymmdd10.))||' '||strip(put(asttm,time5.))||' ('||strip(put(adys{ii},best8.))||')' ;
else dates_c{ii} = strip(put(daten{ii},yymmdd10.))||' ('||strip(put(adys{ii},best8.))||')' ;
end ;
end ;
if missing(endate_c) then
date_c = strip(stdate_c)||' /^nOngoing' ;
else date_c = strip(stdate_c)||' /^n'||strip(endate_c) ;
/* AE SOC, MeDDRA Term, Verbatim */
if aedecod ^= '' then
aecat_c = strip(aebodsys)|| '^n' ||'^R"\b "' ||trim(aedecod) || '^R"\b0 "' || '^n' ||
'^R"\i "' || trim(aeterm) || '^R"\i0"';
else if aebodsys = '' and aedecod= ' ' and aeterm^='' then
aecat_c = '^R"\b "NOT CODED^R"\b0 "' ||'^n' || '^R"\i "' || trim(aeterm) || '^R"\i0"';

/* AE Status flags */
aestat_c = put(trtemfl, $yn.)||'/^n'||put(aeser, $yn.)||'/^n'||put(aesifl, $yn.);
aestat_c = compress(aestat_c, ' ') ;

;

/* AE Severity, Relation to Study Drug Treatment, Outcome */
aesevrel_c = strip(put(atoxgr,$grade.)) || '/^n^R"\b "'||strip(propcase(aerel))||'^R"\b0 "/^n(' ||strip(put(aeout,$aeout.)) || ')'

if aeacn ^='' then aeacn_c = strip(propcase(aeacn)) ;
keep subjid aeseq subjid_c date_c aecat_c aestat_c aesevrel_c aeacn_c;
run ;
%*-----------------------------------------------------------------------------;
%* Indicate New or Changed Data for Listing
%*-----------------------------------------------------------------------------;
%TrackCHG (
indata = afile ,
baseprog = l_ae ,
order
= _samp ,
sortvar = subjid aeseq subjid_c date_c aecat_c aestat_c aesevrel_c aeacn_c ,
arrayvar = subjid_c date_c aecat_c aestat_c aesevrel_c aeacn_c
) ;
%*-----------------------------------------------------------------------------;
%* Generate Listing
%*-----------------------------------------------------------------------------;
ods _all_ close ;
options orientation=landscape ;
ods escapechar='^' ;
title1
j=c "SAMPLE AE LISTING USING TrackCHG MACRO" ;
title2
j=l "Company Co." j=r "Study 001" ;
title3
j=c "Listing 1: Treatment-Emergent Adverse Events" ;
title4
j=c "Safety Population" ;
footnote1 j=l "SAE = serious adverse event;
TEAE = treatment-emergent event is defined as an event occurring during or after
administration of the first dose of study drug until xx days after the final dose
of study drug (follow-up visit)" ;
footnote2 j=l "Note: Adverse event terms were mapped according to Medical
Dictionary for Regulatory Activities (MedDRA) version xx.x." ;
footnote3 j=l "^{super a} Study Day = onset date - first dose date + 1,
if on or after the first study drug dosing; Study Day = onset date - first dose
date, otherwise." ;
ods rtf file="&outputs\listing-01-00-00-l-ae_samp.rtf" style=RTF;
proc report data = _l_ae_samp /*afile*/ missing split = '~' spacing=1 headskip headline spanrows ;
column (subjid aeseq subjid_c date_c aecat_c aestat_c aesevrel_c aeacn_c) ;
define subjid
define aeseq
define subjid_c

/ order order = internal noprint ;
/ order order = internal noprint ;
/ order order = internal 'Subject~ID'

center style=[width=0.65 in];

define date_c
/ display "Onset Date Time (Study Day^{super a})/^nEnd Date Time (Study Day)"
center style=[width=1.75in] ;
define aecat_c
/ display "System Organ Class/~Preferred Term/~^S={font_style=italic}Verbatim Term"
left
style=[width=1.85 in] ;
define aestat_c
/ display "TEAE/~SAE/~Special Interest"
center style=[width=0.85 in] ;
define aesevrel_c
/ display "Severity/~Relationship to Study Drug/~(Outcome)"
left
style=[width=2.03 in] ;
define aeacn_c
/ display "Action Taken with~Study Drug"
left
style=[width=1.85 in] ;
run ;
ods _all_ close ;

